```
In[60]:= Off[ClebschGordan::"tri"];
Off[SixJSymbol::"tri"];
(*PRD32,189-231(1985) Mesons in a relativized quark model with chromodynamics*)
(*计算所用参数*)
α = {0.25, 0.15, 0.20}; γ = {1/2, √10 / 2, √1000 / 2};
mu = 0.22;
ms = 0.419;
mc = 1.628;
mb = 4.977;
mt = 35;
bconst = 0.18;
cconst = -0.253;
σ0 = 1.8;
s = 1.55;
ϵc = -0.168;
ϵt = 0.025;
ϵsov = -0.035;
ϵsos = 0.055;
(*默认参数*)
r1 = 0.5; rmax = 15; nmax = 10;
(*自旋,轨道,张量项,smearing参数σ,τ*)
dspin[S_] := 1/2 S (S + 1) - 3/4;
dso[S_, L_, J_] := J (J + 1) - L (L + 1) - S (S + 1);
dsqq[S_, L_, J_] :=
  If[S == 0 || L == 0, 0, Which[J == L + 1, (J - 1) / 2, J == L, -1/2, J == L - 1, (-J - 2) / 2]];
σij[m1_, m2_] := Sqrt[σ0^2 * (1/2 + 1/2 (4 m1 * m2/(m1 + m2)^2)^4) + s^2 (2 m1 * m2/(m1 + m2))^2];
τij[m1_, m2_, k_] := Sqrt[γ[[k]]^2 * σij[m1, m2]^2/(γ[[k]]^2 + σij[m1, m2]^2)];
tensor[S_, L_, J_] :=
  If[S == 0 || L == 0, 0, Which[J == L + 1, -2 (J - 1)/(2 J + 1), J == L, 2, J == L - 1, -2 (J + 2)/(2 J + 1)]];
Gtilde[m1_, m2_, r_] := -Sum[4 * α[[k]]/(3 r) Erf[τij[m1, m2, k] * r], {k, 1, 3}];
Stilde[m1_, m2_, r_] := bconst * r *
   (Exp[-σij[m1, m2]^2 r^2]/(√π σij[m1, m2] * r) + (1 + 1/(2 σij[m1, m2]^2 * r^2)) Erf[σij[m1, m2] * r]) + cconst;
(*高斯基坐标和动量空间波函数*)
Nnl[i_, l_] := (2^(l+2) (2 νn[i])^(l+3/2)/(√π (2 l + 1)!!))^(1/2);
ϕr[i_, l_, r_] := Nnl[i, l] * r^l * ⅇ^(-νn[i]*r^2);
ϕp[i_, l_, p_] := Nnl[i, l] * 1/(2 νn[i])^(l+3/2) ⅇ^(-p^2/(4 νn[i])) p^l;
(*计算势能项相关矩阵元,动量空间和坐标空间分开计算*)
pij[i_, j_, l_, m1_, m2_, ϵx_] :=
```

```
  NIntegrate[ϕp[i, l, p] * (m1 * m2 / (Sqrt[p^2 + m1^2] Sqrt[p^2 + m2^2]))^(1/2+ϵx) * ϕp[j, l, p] * p^2, {p, 0, ∞}];
Gpij[i_, j_, l_, m1_, m2_] := NIntegrate[
   ϕp[i, l, p] * (1 + p^2 / (Sqrt[p^2 + m1^2] Sqrt[p^2 + m2^2]))^(1/2) * ϕp[j, l, p] * p^2, {p, 0, ∞}];
Gtildeij[i_, j_, l_, m1_, m2_] := NIntegrate[
   ϕr[i, l, r] * Gtilde[m1, m2, r] * ϕr[j, l, r] * r^2, {r, 0, ∞}];
Stildeij[i_, j_, l_, m1_, m2_] := NIntegrate[
   ϕr[i, l, r] * Stilde[m1, m2, r] * ϕr[j, l, r] * r^2, {r, 0, ∞}];
Gij1[i_, j_, l_, m1_, m2_] := NIntegrate[ϕr[i, l, r] * 1/r *
    (D[Gtilde[m1, m2, rp], rp] /. rp → r) * ϕr[j, l, r] * r^2, {r, 0, ∞}];
Gij2[i_, j_, l_, m1_, m2_] := NIntegrate[ϕr[i, l, r] * 1/r^2
    (D[rp^2 * D[Gtilde[m1, m2, rp], rp], rp] /. rp → r) * ϕr[j, l, r] * r^2, {r, 0, ∞}];
Gij3[i_, j_, l_, m1_, m2_] := NIntegrate[ϕr[i, l, r] *
    (D[Gtilde[m1, m2, rp], {rp, 2}] /. rp → r) * ϕr[j, l, r] * r^2, {r, 0, ∞}];
Sij[i_, j_, l_, m1_, m2_] := NIntegrate[ϕr[i, l, r] * 1/r *
    (D[Stilde[m1, m2, rp], rp] /. rp → r) * ϕr[j, l, r] * r^2, {r, 0, ∞}];

mon[i_, j_, l_, m1_, m2_] :=
  NIntegrate[ϕp[i, l, p] * (Sqrt[p^2 + m1^2] + Sqrt[p^2 + m2^2]) * ϕp[j, l, p] * p^2, {p, 0, ∞}];

Eij[i_, j_, l_] := (2 Sqrt[νn[i] × νn[j]] / (νn[i] + νn[j]))^(l+3/2);
(*求解本征方程:Sol函数,返回本征值.
*)
Sol[n1_, l1_, S1_, J1_, m11_, m21_, r11_, rmax1_, nmax1_, zz1_] :=
  Module[{n = n1, l = l11, S = S1, J = J1, m1 = m11, m2 = m21, r1 = r11,
    rmax = rmax1, nmax = nmax1, zz = zz1}, acon = (rmax / r1)^(1/(nmax-1));

   νn[i_] := 1 / (r1 * acon^(i-1))^2;
    (*势能项矩阵*)
   mat1 = Table[pij[i, j, l, m1, m1, ϵsov], {i, 1, nmax}, {j, 1, nmax}];
   mat2 = Table[pij[i, j, l, m2, m2, ϵsov], {i, 1, nmax}, {j, 1, nmax}];
   mat3 = Table[pij[i, j, l, m1, m2, ϵsov], {i, 1, nmax}, {j, 1, nmax}];
   mat4 = Table[pij[i, j, l, m1, m2, ϵc], {i, 1, nmax}, {j, 1, nmax}];
   mat5 = Table[pij[i, j, l, m1, m2, ϵt], {i, 1, nmax}, {j, 1, nmax}];
   mat6 = Table[pij[i, j, l, m1, m1, ϵsos], {i, 1, nmax}, {j, 1, nmax}];
   mat7 = Table[pij[i, j, l, m2, m2, ϵsos], {i, 1, nmax}, {j, 1, nmax}];
   mat8 = Table[Gpij[i, j, l, m1, m2], {i, 1, nmax}, {j, 1, nmax}];
   mat9 = Table[Gtildeij[i, j, l, m1, m2], {i, 1, nmax}, {j, 1, nmax}];
   mat10 = Table[Stildeij[i, j, l, m1, m2], {i, 1, nmax}, {j, 1, nmax}];
   mat11 = Table[Gij1[i, j, l, m1, m2], {i, 1, nmax}, {j, 1, nmax}];
   mat12 = Table[Gij2[i, j, l, m1, m2], {i, 1, nmax}, {j, 1, nmax}];
   mat13 = Table[Gij3[i, j, l, m1, m2], {i, 1, nmax}, {j, 1, nmax}];
```

```
mat14 = Table[Sij[i, j, l, m1, m2], {i, 1, nmax}, {j, 1, nmax}];
(*动能矩阵*)
tt = Table[mon[i, j, l, m1, m2], {i, 1, nmax}, {j, 1, nmax}];
te = Table[Eij[i, j, l], {i, 1, nmax}, {j, 1, nmax}];
tein = Inverse[te];
(*H=T+V*)
th = tt + mat8.tein.mat9.tein.mat8 + dsqq[S, l, J]/(2 m1^2) * mat1.tein.mat11.tein.mat1 +
  dsqq[S, l, J]/(2 m2^2) * mat2.tein.mat11.tein.mat2 + 2 dsqq[S, l, J]/(m1 * m2) *
   mat3.tein.mat11.tein.mat3 + 2 dspin[S]/(3 m1 * m2) * mat4.tein.mat12.tein.mat4 -
  1/12 * tensor[S, l, J]/(m1 * m2) (mat5.tein.mat13.tein.mat5 - mat5.tein.mat11.tein.mat5) +
  mat10 - (dsqq[S, l, J]/(2 m1^2) * mat6.tein.mat14.tein.mat6 +
     dsqq[S, l, J]/(2 m2^2) * mat7.tein.mat14.tein.mat7);
(*求解本征值和本征向量,得到波函数*)
{Ed, wave} = Eigensystem[{th, te}];
en = Ed[[-(n + 1)]];
tb = Table[Sqrt[2 vn[i]], {i, nmax}];
coabc = Sign[Sum[wave[[-(n + 1)]][[i]] * ϕr[i, l, 0.1], {i, 1, nmax}]] *
   (wave[[-(n + 1)]].te.wave[[-(n + 1)]])^(-1/2) * wave[[-(n + 1)]];
unr[n_, l_, r_] := Sum[coabc[[i]] * ϕr[i, l, r] * r, {i, 1, nmax}];

(*输出能量,画图*)
If[zz == 0, Print["Eigenvalue = ", en];
 Print[Plot[unr[n, l, r], {r, 0, 10}]]]; Return[unr[n, l, r]]];
```

Results

```
In[131]:= << MaTeX`
```

J/ψ

```
In[153]:= n1 = 0; l1 = 0; S1 = 1; J1 = 0; m11 = mc; m21 = mc;
psi[r_] = Sol[n1, l1, S1, J1, m11, m21, r1, rmax, nmax, 0];
```

```
Eigenvalue = 3.09138
```



ψ(2S)

In[149]:=
```
n1 = 1; l1 = 0; S1 = 1; J1 = 0; m11 = mc; m21 = mc;
psi2[r_] = Sol[n1, l1, S1, J1, m11, m21, r1, rmax, nmax, 0];
```

```
Eigenvalue = 3.67913
```



Wavefuction

In[155]:=
```
Plot[{psi[r/0.197], psi2[r/0.197]}, {r, 0, 2}, Frame → True,
 PlotTheme → "Scientific", FrameLabel → {MaTeX["r({\\rm fm})"], MaTeX["\\psi(r)"]},
 PlotLegends → {MaTeX["J/\\psi"], MaTeX["\\psi(2S)"]}]
```

Out[155]=


Check Orthonormal

In[159]:=
```
NIntegrate[psi[r]^2, {r, 0, Infinity}]
NIntegrate[psi2[r]^2, {r, 0, Infinity}]
NIntegrate[psi[r] * psi2[r], {r, 0, Infinity}] // Quiet
```

Out[159]= 1.

Out[160]= 1.

Out[161]= 1.3522×10^{-13}

Momentum emission

Naively thinking, $\pi\pi$ emission with 3-momentum q from charmonia can be simply treated as the momentum loss of one c quark. The coupling of $\psi\psi\pi\pi$ should be proportional to the inner production of the initial and final states of charmonia. After the loss of momentum q, the state becomes $|\psi_f> \, = e^{i\,q\,r}\,|\psi_i>$ and then the coupling $\propto \; <\psi_f\,\big|\,\psi_i \geq \int d\,r\,\psi_f(r)^*\,\psi_i(r)\,\frac{\sin(q\,r)}{q\,r}$

In[121]:=
```
c12[q_] := NIntegrate[psi[r] × psi2[r] * Sin[q r]/(q r), {r, 0, 20}]

c11[q_] := NIntegrate[psi[r] × psi[r] * Sin[q r]/(q r), {r, 0, 20}]

c22[q_] := NIntegrate[psi2[r] × psi2[r] * Sin[q r]/(q r), {r, 0, 20}]
```

In[162]:=
```
Plot[{c11[q], c22[q], c12[q]}, {q, 0.01, 1}, Frame → True,
 FrameLabel → {MaTeX["q(\\rm GeV)"]}, PlotLegends → {"c11", "c22", "c12"}]
```

Out[162]=



In[127]:=
```
LogPlot[c11[r] * c22[r] / c12[r]^2, {r, 1, 2}]
```

Out[127]=



In[128]:=
```
LogPlot[c11[r] / c12[r], {r, 2 * 0.138, 1}]
```

Out[128]=